

Optimizing DiffEq Code

Chris Rackauckas

June 27, 2019

In this notebook we will walk through some of the main tools for optimizing your code in order to efficiently solve DifferentialEquations.jl. User-side optimizations are important because, for sufficiently difficult problems, most of the time will be spent inside of your `f` function, the function you are trying to solve. "Efficient" integrators are those that reduce the required number of `f` calls to hit the error tolerance. The main ideas for optimizing your DiffEq code, or any Julia function, are the following:

- Make it non-allocating
- Use StaticArrays for small arrays
- Use broadcast fusion
- Make it type-stable
- Reduce redundant calculations
- Make use of BLAS calls
- Optimize algorithm choice

We'll discuss these strategies in the context of small and large systems. Let's start with small systems.

0.1 Optimizing Small Systems (<100 DEs)

Let's take the classic Lorenz system from before. Let's start by naively writing the system in its out-of-place form:

```
function lorenz(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    [dx,dy,dz]
end
```

```
lorenz (generic function with 1 method)
```

Here, `lorenz` returns an object, `[dx,dy,dz]`, which is created within the body of `lorenz`.

This is a common code pattern from high-level languages like MATLAB, SciPy, or R's `deSolve`. However, the issue with this form is that it allocates a vector, `[dx,dy,dz]`, at each step. Let's benchmark the solution process with this choice of function:

```
using DifferentialEquations, BenchmarkTools
u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 10.63 MiB
  allocs estimate: 99631
  -----
  minimum time:      3.286 ms (0.00% GC)
  median time:       3.893 ms (0.00% GC)
  mean time:         8.375 ms (55.98% GC)
  maximum time:     14.351 ms (71.89% GC)
  -----
  samples:           597
  evals/sample:      1
```

The BenchmarkTools package's `@benchmark` runs the code multiple times to get an accurate measurement. The minimum time is the time it takes when your OS and other background processes aren't getting in the way. Notice that in this case it takes about 5ms to solve and allocates around 11.11 MiB. However, if we were to use this inside of a real user code we'd see a lot of time spent doing garbage collection (GC) to clean up all of the arrays we made. Even if we turn off saving we have these allocations.

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 9.30 MiB
  allocs estimate: 87129
  -----
  minimum time:      2.869 ms (0.00% GC)
  median time:       3.051 ms (0.00% GC)
  mean time:         7.112 ms (55.15% GC)
  maximum time:     13.710 ms (73.50% GC)
  -----
  samples:           703
  evals/sample:      1
```

The problem of course is that arrays are created every time our derivative function is called. This function is called multiple times per step and is thus the main source of memory usage. To fix this, we can use the in-place form to `***make our code non-allocating***`:

```
function lorenz!(du,u,p,t)
  du[1] = 10.0*(u[2]-u[1])
```

```

du[2] = u[1]*(28.0-u[3]) - u[2]
du[3] = u[1]*u[2] - (8/3)*u[3]
end

```

```
lorenz! (generic function with 1 method)
```

Here, instead of creating an array each time, we utilized the cache array `du`. When the inplace form is used, `DifferentialEquations.jl` takes a different internal route that minimizes the internal allocations as well. When we benchmark this function, we will see quite a difference.

```

u0 = [1.0;0.0;0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate:  1.34 MiB
  allocs estimate:  12593
  -----
  minimum time:     875.483 μs (0.00% GC)
  median time:      898.919 μs (0.00% GC)
  mean time:        1.433 ms (36.00% GC)
  maximum time:     10.969 ms (90.53% GC)
  -----
  samples:          3480
  evals/sample:     1

```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```

BenchmarkTools.Trial:
  memory estimate:  6.86 KiB
  allocs estimate:  91
  -----
  minimum time:     472.793 μs (0.00% GC)
  median time:      477.127 μs (0.00% GC)
  mean time:        481.245 μs (0.65% GC)
  maximum time:     10.935 ms (95.10% GC)
  -----
  samples:          10000
  evals/sample:     1

```

There is a 4x time difference just from that change! Notice there are still some allocations and this is due to the construction of the integration cache. But this doesn't scale with the problem size:

```

tspan = (0.0,500.0) # 5x longer than before
prob = ODEProblem(lorenz!,u0,tspan)
@benchmark solve(prob,Tsit5(),save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  6.86 KiB
  allocs estimate:  91
  -----
  minimum time:     2.440 ms (0.00% GC)
  median time:      2.455 ms (0.00% GC)
  mean time:        2.462 ms (0.00% GC)
  maximum time:     2.638 ms (0.00% GC)
  -----
  samples:          2030
  evals/sample:     1

```

since that's all just setup allocations.

But if the system is small we can optimize even more. Allocations are only expensive if they are "heap allocations". For a more in-depth definition of heap allocations, [there are a lot of sources online](#). But a good working definition is that heap allocations are variable-sized slabs of memory which have to be pointed to, and this pointer indirection costs time. Additionally, the heap has to be managed and the garbage controllers has to actively keep track of what's on the heap.

However, there's an alternative to heap allocations, known as stack allocations. The stack is statically-sized (known at compile time) and thus its accesses are quick. Additionally, the exact block of memory is known in advance by the compiler, and thus re-using the memory is cheap. This means that allocating on the stack has essentially no cost!

Arrays have to be heap allocated because their size (and thus the amount of memory they take up) is determined at runtime. But there are structures in Julia which are stack-allocated. **structs** for example are stack-allocated "value-type"s. **Tuples** are a stack-allocated collection. The most useful data structure for DiffEq though is the **StaticArray** from the package [StaticArrays.jl](#). These arrays have their length determined at compile-time. They are created using macros attached to normal array expressions, for example:

```

using StaticArrays
A = @SVector [2.0,3.0,5.0]

```

```

3-element StaticArrays.SArray{Tuple{3},Float64,1,3}:
 2.0
 3.0
 5.0

```

Notice that the 3 after **SVector** gives the size of the **SVector**. It cannot be changed. Additionally, **SVectors** are immutable, so we have to create a new **SVector** to change values. But remember, we don't have to worry about allocations because this data structure is stack-allocated. **SArrays** have a lot of extra optimizations as well: they have fast matrix multiplication, fast QR factorizations, etc. which directly make use of the information about the size of the array. Thus, when possible they should be used.

Unfortunately static arrays can only be used for sufficiently small arrays. After a certain size, they are forced to heap allocate after some instructions and their compile time balloons.

Thus static arrays shouldn't be used if your system has more than 100 variables. Additionally, only the native Julia algorithms can fully utilize static arrays.

Let's *****optimize `lorenz` using static arrays*****. Note that in this case, we want to use the out-of-place allocating form, but this time we want to output a static array:

```
function lorenz_static(u,p,t)
    dx = 10.0*(u[2]-u[1])
    dy = u[1]*(28.0-u[3]) - u[2]
    dz = u[1]*u[2] - (8/3)*u[3]
    @SVector [dx,dy,dz]
end
```

```
lorenz_static (generic function with 1 method)
```

To make the solver internally use static arrays, we simply give it a static array as the initial condition:

```
u0 = @SVector [1.0,0.0,0.0]
tspan = (0.0,100.0)
prob = ODEProblem(lorenz_static,u0,tspan)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 461.59 KiB
  allocs estimate: 2583
  -----
  minimum time:      497.357 μs (0.00% GC)
  median time:       514.542 μs (0.00% GC)
  mean time:         1.398 ms (9.68% GC)
  maximum time:      53.840 ms (0.00% GC)
  -----
  samples:           3568
  evals/sample:      1
```

```
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 6.16 KiB
  allocs estimate: 73
  -----
  minimum time:      399.061 μs (0.00% GC)
  median time:       403.103 μs (0.00% GC)
  mean time:         407.300 μs (0.47% GC)
  maximum time:      10.052 ms (95.59% GC)
  -----
  samples:           10000
  evals/sample:      1
```

And that's pretty much all there is to it. With static arrays you don't have to worry about allocating, so use operations like `*` and don't worry about fusing operations (discussed in the next section). Do "the vectorized code" of R/MATLAB/Python and your code in this case will be fast, or directly use the numbers/values.

Exercise 1 Implement the out-of-place array, in-place array, and out-of-place static array forms for the [Henon-Heiles System](#) and time the results.

0.2 Optimizing Large Systems

0.2.1 Interlude: Managing Allocations with Broadcast Fusion

When your system is sufficiently large, or you have to make use of a non-native Julia algorithm, you have to make use of `Arrays`. In order to use arrays in the most efficient manner, you need to be careful about temporary allocations. Vectorized calculations naturally have plenty of temporary array allocations. This is because a vectorized calculation outputs a vector. Thus:

```
A = rand(1000,1000); B = rand(1000,1000); C = rand(1000,1000)
test(A,B,C) = A + B + C
@benchmark test(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  3
  -----
  minimum time:     3.311 ms (0.00% GC)
  median time:      3.364 ms (0.00% GC)
  mean time:        4.959 ms (32.63% GC)
  maximum time:     8.497 ms (57.27% GC)
  -----
  samples:          1006
  evals/sample:     1
```

That expression `A + B + C` creates 2 arrays. It first creates one for the output of `A + B`, then uses that result array to `+` `C` to get the final result. 2 arrays! We don't want that! The first thing to do to fix this is to use broadcast fusion. [Broadcast fusion](#) puts expressions together. For example, instead of doing the `+` operations separately, if we were to add them all at the same time, then we would only have a single array that's created. For example:

```
test2(A,B,C) = map((a,b,c)->a+b+c,A,B,C)
@benchmark test2(A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  5
  -----
  minimum time:     3.363 ms (0.00% GC)
```

```

median time:      3.385 ms (0.00% GC)
mean time:        5.014 ms (32.30% GC)
maximum time:     8.611 ms (59.69% GC)
-----
samples:          995
evals/sample:     1

```

Puts the whole expression into a single function call, and thus only one array is required to store output. This is the same as writing the loop:

```

function test3(A,B,C)
    D = similar(A)
    @inbounds for i in eachindex(A)
        D[i] = A[i] + B[i] + C[i]
    end
    D
end
@benchmark test3(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:     3.292 ms (0.00% GC)
median time:      3.343 ms (0.00% GC)
mean time:        4.954 ms (32.77% GC)
maximum time:     9.503 ms (64.80% GC)
-----
samples:          1007
evals/sample:     1

```

However, Julia's broadcast is syntactic sugar for this. If multiple expressions have a `.`, then it will put those vectorized operations together. Thus:

```

test4(A,B,C) = A .+ B .+ C
@benchmark test4(A,B,C)

```

```

BenchmarkTools.Trial:
 memory estimate:  7.63 MiB
 allocs estimate:  2
-----
minimum time:     3.311 ms (0.00% GC)
median time:      3.338 ms (0.00% GC)
mean time:        4.965 ms (32.88% GC)
maximum time:     8.905 ms (60.60% GC)
-----
samples:          1005
evals/sample:     1

```

is a version with only 1 array created (the output). Note that `.`s can be used with function calls as well:

```
sin.(A) .+ sin.(B)
```

```
1000×1000 Array{Float64,2}:
```

```
 0.931108  0.770393  0.789677  0.218802  ...  0.22953  1.51211  1.00179
 0.90054   0.602936  0.725035  0.790666      0.810387  0.744    0.94451
 0.715065  0.480829  0.575951  0.72759      1.11909  1.52666  0.280234
 1.39019   0.320839  1.33148   0.513073      0.357953  0.894232  0.84505
 0.604789  1.29015   0.852567  0.428484      0.570895  0.957067  0.578171
 0.868696  1.09534   0.27991   1.44741  ...  1.11781  0.884421  0.827647
 0.997828  0.821506  0.760147  1.10463      0.955639  1.30173  0.832633
 0.653481  1.2737    1.14446   0.941175      0.597795  0.854467  1.09635
 0.716143  0.859716  1.45342   0.906673      0.592314  1.42853  0.982148
 0.467133  0.495139  0.852799  0.601891      0.872183  1.12925  1.52778
 ⋮
 0.577652  0.287797  0.653243  0.548965      1.0065    0.72673  0.884134
 0.852376  1.31456   1.33269   1.3329      0.858516  1.01696  0.739127
 0.457484  0.735204   1.42047   0.621278      1.12957  0.484122  0.706761
 0.799339  0.56432    0.934546  0.484977      0.422371  1.41029  0.609449
 0.89385   1.57318    0.771652  0.0525704 ...  0.879817  0.825414  0.754669
 1.43111   0.845991   1.15616   1.14194      1.39905  0.712747  1.27942
 1.56422   0.91982    0.909935  0.963826      0.645371  0.873904  1.22899
 1.19556   0.888317   1.18897   0.827397      0.817037  1.04366  1.57913
 0.604812  0.87215    1.07305   0.373738      0.855046  0.42142  1.4774
```

Also, the @. macro applies a dot to every operator:

```
test5(A,B,C) = @. A + B + C #only one array allocated
@benchmark test5(A,B,C)
```

```
BenchmarkTools.Trial:
```

```
memory estimate: 7.63 MiB
```

```
allocs estimate: 3
```

```
-----
```

```
minimum time:      3.318 ms (0.00% GC)
```

```
median time:       3.365 ms (0.00% GC)
```

```
mean time:         4.972 ms (32.83% GC)
```

```
maximum time:      8.370 ms (60.01% GC)
```

```
-----
```

```
samples:           1003
```

```
evals/sample:      1
```

Using these tools we can get rid of our intermediate array allocations for many vectorized function calls. But we are still allocating the output array. To get rid of that allocation, we can instead use mutation. Mutating broadcast is done via .=. For example, if we pre-allocate the output:

```
D = zeros(1000,1000);
```

Then we can keep re-using this cache for subsequent calculations. The mutating broadcasting form is:


```
test6!(D,A,B,C) = D .= A .+ B .+ C #only one array allocated
@benchmark test6!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 0 bytes
  allocs estimate: 0
  -----
  minimum time:      3.263 ms (0.00% GC)
  median time:       3.283 ms (0.00% GC)
  mean time:         3.288 ms (0.00% GC)
  maximum time:      3.624 ms (0.00% GC)
  -----
  samples:           1514
  evals/sample:      1
```

If we use `@.` before the `=`, then it will turn it into `.=`:

```
test7!(D,A,B,C) = @. D = A + B + C #only one array allocated
@benchmark test7!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 0 bytes
  allocs estimate: 0
  -----
  minimum time:      3.264 ms (0.00% GC)
  median time:       3.286 ms (0.00% GC)
  mean time:         3.291 ms (0.00% GC)
  maximum time:      3.527 ms (0.00% GC)
  -----
  samples:           1513
  evals/sample:      1
```

Notice that in this case, there is no "output", and instead the values inside of `D` are what are changed (like with the `DiffEq` inplace function). Many Julia functions have a mutating form which is denoted with a `!`. For example, the mutating form of the `map` is `map!`:

```
test8!(D,A,B,C) = map!((a,b,c)->a+b+c,D,A,B,C)
@benchmark test8!(D,A,B,C)
```

```
BenchmarkTools.Trial:
  memory estimate: 32 bytes
  allocs estimate: 1
  -----
  minimum time:      3.263 ms (0.00% GC)
  median time:       3.310 ms (0.00% GC)
  mean time:         3.317 ms (0.00% GC)
  maximum time:      3.718 ms (0.00% GC)
  -----
  samples:           1502
  evals/sample:      1
```

Some operations require using an alternate mutating form in order to be fast. For example, matrix multiplication via `*` allocates a temporary:

```
@benchmark A*B
```

```
BenchmarkTools.Trial:
  memory estimate:  7.63 MiB
  allocs estimate:  2
  -----
  minimum time:      21.824 ms (0.00% GC)
  median time:       21.953 ms (0.00% GC)
  mean time:         23.660 ms (7.02% GC)
  maximum time:      29.881 ms (0.00% GC)
  -----
  samples:           212
  evals/sample:      1
```

Instead, we can use the mutating form `mul!` into a cache array to avoid allocating the output:

```
using LinearAlgebra
@benchmark mul!(D,A,B) # same as D = A * B
```

```
BenchmarkTools.Trial:
  memory estimate:  0 bytes
  allocs estimate:  0
  -----
  minimum time:      21.243 ms (0.00% GC)
  median time:       21.340 ms (0.00% GC)
  mean time:         21.342 ms (0.00% GC)
  maximum time:      22.146 ms (0.00% GC)
  -----
  samples:           235
  evals/sample:      1
```

For repeated calculations this reduced allocation can stop GC cycles and thus lead to more efficient code. Additionally, ***we can fuse together higher level linear algebra operations using BLAS***. The package [SugarBLAS.jl](#) makes it easy to write higher level operations like `alpha*B*A + beta*C` as mutating BLAS calls.

0.2.2 Example Optimization: Gierer-Meinhardt Reaction-Diffusion PDE Discretization

Let's optimize the solution of a Reaction-Diffusion PDE's discretization. In its discretized form, this is the ODE:

$$du = D_1(A_y u + u A_x) + \frac{au^2}{v} + \bar{u} - \alpha u \quad (1)$$

$$dv = D_2(A_y v + v A_x) + au^2 + \beta v \quad (2)$$

where u , v , and A are matrices. Here, we will use the simplified version where A is the tridiagonal stencil $[1, -2, 1]$, i.e. it's the 2D discretization of the Laplacian. The native code would be something along the lines of:

```
# Generate the constants
p = (1.0,1.0,1.0,10.0,0.001,100.0) # a,α,ubar,β,D1,D2
N = 100
Ax = Array{Tridiagonal{Float64,3}}{1.0 for i in 1:N-1, [-2.0 for i in 1:N], [1.0 for i in 1:N-1]]
Ay = copy{Tridiagonal{Float64,3}}(Ax)
Ax[2,1] = 2.0
Ax[end-1,end] = 2.0
Ay[1,2] = 2.0
Ay[end,end-1] = 2.0

function basic_version!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = r[:, :, 1]
    v = r[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    dr[:, :, 1] = Du ./ a + u.*u./v + ubar - α*u
    dr[:, :, 2] = Dv ./ a + u.*u - β*v
end

a,α,ubar,β,D1,D2 = p
uss = (ubar+β)/α
vss = (a/β)*uss^2
r0 = zeros{Float64,3}(100,100,2)
r0[:, :, 1] .= uss + 0.1.*rand{Float64}()
r0[:, :, 2] .= vss

prob = ODEProblem{Float64,3}(basic_version!,r0,(0.0,0.1),p)
```

```
ODEProblem with uType Array{Float64,3} and tType Float64. In-place: true
timespan: (0.0, 0.1)
u0: [11.0703 11.0794 ... 11.0648 11.0288; 11.0584 11.0269 ... 11.0766 11.0163;
... ; 11.0636 11.0231 ... 11.0288 11.0432; 11.0943 11.0362 ... 11.0124 11.0528]

[12.1 12.1 ... 12.1 12.1; 12.1 12.1 ... 12.1 12.1; ... ; 12.1 12.1 ... 12.1 12.1; 1
2.1 12.1 ... 12.1 12.1]
```

In this version we have encoded our initial condition to be a 3-dimensional array, with $u[:, :, 1]$ being the A part and $u[:, :, 2]$ being the B part.

```
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 186.88 MiB
  allocs estimate: 8589
  -----
  minimum time:      7.118 s (0.72% GC)
  median time:       7.118 s (0.72% GC)
  mean time:         7.118 s (0.72% GC)
```

```

maximum time:      7.118 s (0.72% GC)
-----
samples:          1
evals/sample:     1

```

While this version isn't very efficient,

We recommend writing the "high-level" code first, and iteratively optimizing it! The first thing that we can do is get rid of the slicing allocations. The operation `r[:, :, 1]` creates a temporary array instead of a "view", i.e. a pointer to the already existing memory. To make it a view, add `@view`. Note that we have to be careful with views because they point to the same memory, and thus changing a view changes the original values:

```

A = rand(4)
@show A

```

```

A = [0.656811, 0.772796, 0.0666667, 0.177342]

```

```

B = @view A[1:3]
B[2] = 2
@show A

```

```

A = [0.656811, 2.0, 0.0666667, 0.177342]
4-element Array{Float64,1}:
 0.6568112064528353
 2.0
 0.06666674805559292
 0.17734186211285796

```

Notice that changing B changed A. This is something to be careful of, but at the same time we want to use this since we want to modify the output `dr`. Additionally, the last statement is a purely element-wise operation, and thus we can make use of broadcast fusion there. Let's rewrite `basic_version!` to `***avoid slicing allocations***` and to `***use broadcast fusion***`:

```

function gm2!(dr,r,p,t)
    a,α,ubar,β,D1,D2 = p
    u = @view r[:, :, 1]
    v = @view r[:, :, 2]
    du = @view dr[:, :, 1]
    dv = @view dr[:, :, 2]
    Du = D1*(Ay*u + u*Ax)
    Dv = D2*(Ay*v + v*Ax)
    @. du = Du + a.*u.*u./v + ubar - α*u
    @. dv = Dv + a.*u.*u - β*v
end
prob = ODEProblem(gm2!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 119.55 MiB
  allocs estimate: 7119
  -----
  minimum time:      4.102 s (0.79% GC)
  median time:       4.677 s (0.70% GC)
  mean time:         4.677 s (0.70% GC)
  maximum time:      5.252 s (0.64% GC)
  -----
  samples:           2
  evals/sample:      1

```

Now, most of the allocations are taking place in $Du = D1*(Ay*u + u*Ax)$ since those operations are vectorized and not mutating. We should instead replace the matrix multiplications with `mul!`. When doing so, we will need to have cache variables to write into. This looks like:

```

Ayu = zeros(N,N)
uAx = zeros(N,N)
Du = zeros(N,N)
Ayv = zeros(N,N)
vAx = zeros(N,N)
Dv = zeros(N,N)
function gm3!(dr,r,p,t)
  a,α,ubar,β,D1,D2 = p
  u = @view r[:, :, 1]
  v = @view r[:, :, 2]
  du = @view dr[:, :, 1]
  dv = @view dr[:, :, 2]
  mul!(Ayu,Ay,u)
  mul!(uAx,u,Ax)
  mul!(Ayv,Ay,v)
  mul!(vAx,v,Ax)
  @. Du = D1*(Ayu + uAx)
  @. Dv = D2*(Ayv + vAx)
  @. du = Du + a*u*u./v + ubar - α*u
  @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm3!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 29.76 MiB
  allocs estimate: 5355
  -----
  minimum time:      69.650 ms (7.35% GC)
  median time:       69.991 ms (7.36% GC)
  mean time:         71.561 ms (9.46% GC)
  maximum time:      75.369 ms (13.63% GC)
  -----
  samples:           70
  evals/sample:      1

```

But our temporary variables are global variables. We need to either declare the caches as `const` or localize them. We can localize them by adding them to the parameters, `p`. It's

easier for the compiler to reason about local variables than global variables. *****Localizing variables helps to ensure type stability***.**

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,Ayu,uAx,Du,Ayv,vAx,Dv) # a,α,ubar,β,D1,D2
function gm4!(dr,r,p,t)
    a,α,ubar,β,D1,D2,Ayu,uAx,Du,Ayv,vAx,Dv = p
    u = @view r[:,1]
    v = @view r[:,2]
    du = @view dr[:,1]
    dv = @view dr[:,2]
    mul!(Ayu,Ay,u)
    mul!(uAx,u,Ax)
    mul!(Ayv,Ay,v)
    mul!(vAx,v,Ax)
    @. Du = D1*(Ayu + uAx)
    @. Dv = D2*(Ayv + vAx)
    @. du = Du + a*u*u./v + ubar - α*u
    @. dv = Dv + a*u*u - β*v
end
prob = ODEProblem(gm4!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())
```

```
BenchmarkTools.Trial:
  memory estimate: 29.66 MiB
  allocs estimate: 1090
  -----
  minimum time:      60.996 ms (8.30% GC)
  median time:       63.302 ms (8.05% GC)
  mean time:         64.163 ms (10.46% GC)
  maximum time:      103.738 ms (9.85% GC)
  -----
  samples:           78
  evals/sample:      1
```

We could then use the BLAS `gemmv` to optimize the matrix multiplications some more, but instead let's devectorize the stencil.

```
p = (1.0,1.0,1.0,10.0,0.001,100.0,N)
function fast_gm!(du,u,p,t)
    a,α,ubar,β,D1,D2,N = p

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
                    a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    end

    @inbounds for j in 2:N-1, i in 2:N-1
        du[i,j,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
                    a*u[i,j,1]^2 - β*u[i,j,2]
    end

    @inbounds for j in 2:N-1
        i = 1
        du[1,j,1] = D1*(2u[i+1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
                    a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
```

```

end
@inbounds for j in 2:N-1
    i = 1
    du[1,j,2] = D2*(2u[i+1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
@inbounds for j in 2:N-1
    i = N
    du[end,j,1] = D1*(2u[i-1,j,1] + u[i,j+1,1] + u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for j in 2:N-1
    i = N
    du[end,j,2] = D2*(2u[i-1,j,2] + u[i,j+1,2] + u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds for i in 2:N-1
    j = 1
    du[i,1,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = 1
    du[i,1,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,1] = D1*(u[i-1,j,1] + u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
end
@inbounds for i in 2:N-1
    j = N
    du[i,end,2] = D2*(u[i-1,j,2] + u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]
end

@inbounds begin
    i = 1; j = 1
    du[1,1,1] = D1*(2u[i+1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,1,2] = D2*(2u[i+1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = 1; j = N
    du[1,N,1] = D1*(2u[i+1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[1,N,2] = D2*(2u[i+1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = 1
    du[N,1,1] = D1*(2u[i-1,j,1] + 2u[i,j+1,1] - 4u[i,j,1]) +
        a*u[i,j,1]^2/u[i,j,2] + ubar - α*u[i,j,1]
    du[N,1,2] = D2*(2u[i-1,j,2] + 2u[i,j+1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 - β*u[i,j,2]

    i = N; j = N
    du[end,end,1] = D1*(2u[i-1,j,1] + 2u[i,j-1,1] - 4u[i,j,1]) +

```

```

        a*u[i,j,1]^2/u[i,j,2] + ubar -  $\alpha$ *u[i,j,1]
    du[end,end,2] = D2*(2u[i-1,j,2] + 2u[i,j-1,2] - 4u[i,j,2]) +
        a*u[i,j,1]^2 -  $\beta$ *u[i,j,2]
end
end
prob = ODEProblem(fast_gm!,r0,(0.0,0.1),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 29.63 MiB
  allocs estimate: 504
  -----
  minimum time:      17.785 ms (26.13% GC)
  median time:       17.940 ms (26.41% GC)
  mean time:         19.415 ms (32.14% GC)
  maximum time:      22.906 ms (43.06% GC)
  -----
  samples:           258
  evals/sample:      1

```

Lastly, we can do other things like multithread the main loops, but these optimizations get the last 2x-3x out. The main optimizations which apply everywhere are the ones we just performed (though the last one only works if your matrix is a stencil. This is known as a matrix-free implementation of the PDE discretization).

This gets us to about 8x faster than our original MATLAB/SciPy/R vectorized style code!

The last thing to do is then *****optimize our algorithm choice*****. We have been using `Tsit5()` as our test algorithm, but in reality this problem is a stiff PDE discretization and thus one recommendation is to use `CVODE_BDF()`. However, instead of using the default dense Jacobian, we should make use of the sparse Jacobian afforded by the problem. The Jacobian is the matrix $\frac{df_i}{dr_j}$, where r is read by the linear index (i.e. down columns). But since the u variables depend on the v , the band size here is large, and thus this will not do well with a Banded Jacobian solver. Instead, we utilize sparse Jacobian algorithms. `CVODE_BDF` allows us to use a sparse Newton-Krylov solver by setting `linear_solver = :GMRES` (see [the solver documentation](#), and thus we can solve this problem efficiently. Let's see how this scales as we increase the integration time.

```

prob = ODEProblem(fast_gm!,r0,(0.0,10.0),p)
@benchmark solve(prob,Tsit5())

```

```

BenchmarkTools.Trial:
  memory estimate: 2.76 GiB
  allocs estimate: 41670
  -----
  minimum time:      58.303 s (7.14% GC)
  median time:       58.303 s (7.14% GC)
  mean time:         58.303 s (7.14% GC)
  maximum time:      58.303 s (7.14% GC)
  -----
  samples:           1
  evals/sample:      1

```



```
using Sundials
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate: 120.14 MiB
  allocs estimate: 31735
  -----
  minimum time:      807.479 ms (3.46% GC)
  median time:      902.584 ms (20.80% GC)
  mean time:        909.848 ms (19.90% GC)
  maximum time:     987.067 ms (27.37% GC)
  -----
  samples:           6
  evals/sample:      1
```

```
prob = ODEProblem(fast_gm!,r0,(0.0,100.0),p)
# Will go out of memory if we don't turn off `save_everystep`!
@benchmark solve(prob,Tsit5(),save_everystep=false)
```

```
BenchmarkTools.Trial:
  memory estimate: 2.91 MiB
  allocs estimate: 112
  -----
  minimum time:      9.408 s (0.00% GC)
  median time:      9.408 s (0.00% GC)
  mean time:        9.408 s (0.00% GC)
  maximum time:     9.408 s (0.00% GC)
  -----
  samples:           1
  evals/sample:      1
```

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES))
```

```
BenchmarkTools.Trial:
  memory estimate: 298.52 MiB
  allocs estimate: 78258
  -----
  minimum time:      2.072 s (9.35% GC)
  median time:      2.102 s (9.22% GC)
  mean time:        2.181 s (10.20% GC)
  maximum time:     2.368 s (19.99% GC)
  -----
  samples:           3
  evals/sample:      1
```

Now let's check the allocation growth.

```
@benchmark solve(prob,CVODE_BDF(linear_solver=:GMRES),save_everystep=false)
```

```

BenchmarkTools.Trial:
  memory estimate:  4.10 MiB
  allocs estimate:  69237
  -----
  minimum time:     1.826 s (0.00% GC)
  median time:      1.827 s (0.00% GC)
  mean time:        1.828 s (0.00% GC)
  maximum time:     1.832 s (0.00% GC)
  -----
  samples:          3
  evals/sample:     1

prob = ODEProblem(fast_gm!, r0, (0.0, 500.0), p)
@benchmark solve(prob, CVODE_BDF(linear_solver=:GMRES), save_everystep=false)

```

```

BenchmarkTools.Trial:
  memory estimate:  5.38 MiB
  allocs estimate:  96997
  -----
  minimum time:     2.563 s (0.00% GC)
  median time:      2.566 s (0.00% GC)
  mean time:        2.566 s (0.00% GC)
  maximum time:     2.569 s (0.00% GC)
  -----
  samples:          2
  evals/sample:     1

```

Notice that we've eliminated almost all allocations, allowing the code to grow without hitting garbage collection and slowing down.

Why is `CVODE_BDF` doing well? What's happening is that, because the problem is stiff, the number of steps required by the explicit Runge-Kutta method grows rapidly, whereas `CVODE_BDF` is taking large steps. Additionally, the `GMRES` linear solver form is quite an efficient way to solve the implicit system in this case. This is problem-dependent, and in many cases using a Krylov method effectively requires a preconditioner, so you need to play around with testing other algorithms and linear solvers to find out what works best with your problem.

0.3 Conclusion

Julia gives you the tools to optimize the solver "all the way", but you need to make use of it. The main thing to avoid is temporary allocations. For small systems, this is effectively done via static arrays. For large systems, this is done via in-place operations and cache arrays. Either way, the resulting solution can be immensely sped up over vectorized formulations by using these principles.

0.4 Appendix

This tutorial is part of the `DiffEqTutorials.jl` repository, found at: <https://github.com/JuliaDiffEq/DiffEqTutorials.jl>

To locally run this tutorial, do the following commands:

```
using DiffEqTutorials
DiffEqTutorials.weave_file("introduction","03-optimizing_diffeq_code.jmd")
```

Computer Information:

```
Julia Version 1.1.1
Commit 55e36cc308 (2019-05-16 04:10 UTC)
Platform Info:
  OS: Linux (x86_64-pc-linux-gnu)
  CPU: Intel(R) Core(TM) i7-3770 CPU @ 3.40GHz
  WORD_SIZE: 64
  LIBM: libopenlibm
  LLVM: libLLVM-6.0.1 (ORCJIT, ivybridge)
```

Package Information:

```
Status `~/julia/environments/v1.1/Project.toml`
[7e558dbc-694d-5a72-987c-6f4ebed21442] ArbNumerics 0.5.4
[6e4b80f9-dd63-53aa-95a3-0cdb28fa8baf] BenchmarkTools 0.4.2
[be33ccc6-a3ff-5ff2-a52e-74243cff1e17] CUDAnative 2.2.0
[3a865a2d-5b23-5a0f-bc46-62713ec82fae] CuArrays 1.0.2
[55939f99-70c6-5e9b-8bb0-5071ed7d61fd] DecFP 0.4.8
[abce61dc-4473-55a0-ba07-351d65e31d42] Decimals 0.4.0
[ebbdde9d-f333-5424-9be2-dbf1e9acfb5e] DiffEqBayes 1.1.0
[eb300fae-53e8-50a0-950c-e21f52c2b7e0] DiffEqBiological 3.8.2
[459566f4-90b8-5000-8ac3-15dfb0a30def] DiffEqCallbacks 2.5.2
[f3b72e0c-5b89-59e1-b016-84e28bfd966d] DiffEqDevTools 2.9.0
[1130ab10-4a5a-5621-a13d-e4788d82bd4c] DiffEqParamEstim 1.6.0
[055956cb-9e8b-5191-98cc-73ae4a59e68a] DiffEqPhysics 3.1.0
[6d1b261a-3be8-11e9-3f2f-0b112a9a8436] DiffEqTutorials 0.1.0
[0c46a032-eb83-5123-abaf-570d42b7fbba] DifferentialEquations 6.4.0
[31c24e10-a181-5473-b8eb-7969acd0382f] Distributions 0.20.0
[497a8b3b-efae-58df-a0af-a86822472b78] DoubleFloats 0.9.1
[f6369f11-7733-5829-9624-2563aa707210] ForwardDiff 0.10.3
[c91e804a-d5a3-530f-b6f0-dfbca275c004] Gadfly 1.0.1
[7073ff75-c697-5162-941a-fcdaad2a7d2a] IJulia 1.18.1
[4138dd39-2aa7-5051-a626-17a0bb65d9c8] JLD 0.9.1
[23fbe1c1-3f47-55db-b15f-69d7ec21a316] Latexify 0.8.2
[eff96d63-e80a-5855-80a2-b1b0885c5ab7] Measurements 2.0.0
[961ee093-0014-501f-94e3-6117800e7a78] ModelingToolkit 0.2.0
[76087f3c-5699-56af-9a33-bf431cd00edd] NLOpt 0.5.1
[2774e3e8-f4cf-5e23-947b-6d7e65073b56] NLSolve 4.0.0
[429524aa-4258-5aef-a3af-852621145aeb] Optim 0.18.1
[1dea7af3-3e70-54e6-95c3-0bf5283fa5ed] OrdinaryDiffEq 5.8.1
[65888b18-ceab-5e60-b2b9-181511a3b968] ParameterizedFunctions 4.1.1
```

[91a5bcdd-55d7-5caf-9e0b-520d859cae80] Plots 0.25.1
[d330b81b-6aea-500a-939a-2ce795aea3ee] PyPlot 2.8.1
[731186ca-8d62-57ce-b412-fbd966d074cd] RecursiveArrayTools 0.20.0
[90137ffa-7385-5640-81b9-e52037218182] StaticArrays 0.11.0
[f3b207a7-027a-5e70-b257-86293d7955fd] StatsPlots 0.11.0
[c3572dad-4567-51f8-b174-8c6c989267f4] Sundials 3.6.1
[1986cc42-f94f-5a68-af5c-568840ba703d] Unitful 0.15.0
[44d3d7a6-8a23-5bf8-98c5-b353f8df5ec9] Weave 0.9.0
[b77e0a4c-d291-57a0-90e8-8db25a27a240] InteractiveUtils
[37e2e46d-f89d-539d-b4ee-838fcccc9c8e] LinearAlgebra
[44cfe95a-1eb2-52ea-b672-e2afdf69b78f] Pkg